Math 1800 Project 4: Stationary Points with Mathematica

Subhadip Chowdhury

A stationary point for a function of two variables is a point where both first partial derivatives equal zero. In this project you will investigate the stationary points for the following functions:

$$f(x,y) = x^3 + 3xy + y^3$$
$$g(x,y) = x^2 + 6xy + y^2 + 14x + 10y$$
$$h(x,y) = 16x^2 + 8xy + y^2$$

Computing Stationary Points

Not only will *Mathematica* calculate the partial derivatives for you, but it also has a built-in `NSolve` command that you can use to find the points where the derivatives are both zero.

(1) Define the first function by executing the command

```
f[x_,y_] := x^3 + 3*x*y + y^3
```

Similarly define $g(x,y)$ and $h(x,y)$ next.

(2) Solve for the stationary points by executing the command

```
Solve[Grad[f[x,y],{x,y}]=={0,0}, {x, y}, Reals]
```

(3) We can define a routine `StatPts` in Mathematica that will take f, g or h as input and produce the list of stationary points directly as follows. Type and execute

```
StatPts[func_] := Solve[Grad[func[x,y], {x, y}] == {0, 0}, {x, y}, Reals]
```

Check that `StatPts[f]` produces the same list of points as part (2). The advantages of doing this step are as follows.

- First, to find the stationary points of `g` and `h`, we can skip writing a long command as in part (2). Instead, we can directly get a list by using `StatPts[g]` and `StatPts[h]`.

- Secondly, we have given the list of stationary points a name, that we can refer to later.

(4) Find and record these stationary points for future reference in the table below. *Note that the third function $h(x,y)$ has an entire line of stationary points, and you should choose any **one** of these for your investigation.*

Question 1

Draw the 3D plots and the contour plots of the functions f, g and h and try to visually classify your critical points as local maximum, local minimum, or saddle point. *Choose a big enough domain so that it contains all the stationary points you are looking at.* Recall that the command for 3D Plot looks like

```
Plot3D[f[x,y],{x,a,b}, {y,c,d}]
```

and the command for Contour Plot looks like

```
ContourPlot[f[x,y],{x,a,b}, {y,c,d}]
```

Note that you will need to replace a, b, c and d with appropriate numbers to see the full pictures.

Question 2

Fill out the following table with information you obtained from above plots.

Function:	$f(x,y)$		$g(x,y)$	$h(x,y)$
Stationary points: $x =$				
$y =$				
Sign of $\frac{\partial^2}{\partial x^2}$:				
Sign of $\frac{\partial^2}{\partial y^2}$:				
Classification:				

Using Second Derivative test

You may have found that it was hard to visually classify the stationary points of g. We can use the second derivative test to give a definite answer in such cases.

(5) Recall that the Hessian is the matrix $\mathbf{H} = \begin{bmatrix} f_{xx} & f_{xy} \\ f_{yx} & f_{yy} \end{bmatrix}$. We can think of this as the gradient of the gradient of f. Type and execute

```
H[func_]:=Grad[Grad[func[x,y],{x,y}],{x,y}]
```

to define a routine `H`. Check that `H[f,{x,y}]` produces the Hessian matrix for f.

(6) We can define the determinant of the Hessian `d` as

```
d[func_]:=Det[H[func]]
```

Execute the command above.

(7) Find `d[f]`. Next find the determinant for other two functions by changing f to g and h.

(8) We can evaluate the determinant at each of the stationary point above as follows. Type and execute

```
d[f]/.StatPts[f]
```

to evaluate $d(f,x,y)$ at the stationary points of f. Recall that the '`/.`' operator was used in last lab, it's called `ReplaceAll`.

(9) You can get the top left entry in the Hessian matrix by typing `H[f][[1,1]]`. Find the value of f_{xx} at the stationary points by executing

```
H[f][[1,1]]/.StatPts[f]
```

Write down the values of f_{xx}, g_{xx}, and h_{xx} at corresponding stationary points from the table. You can copy and paste above lines of codes and change f to g or h everywhere to investigate stationary points of g and h.

Question 3

Classify the stationary points as local maxima, local minima or saddle point using the values you obtained at part (9). Check that your classification is consistent with the table above.

Question 4

Is the following "second derivative test" a valid method?

> Consider a stationary point (a,b) for any function $f(x,y)$.
> If $f_{xx}(a,b) > 0$ and $f_{yy}(a,b) > 0$, then (a,b) is a local minima.
> If $f_{xx}(a,b) < 0$ and $f_{yy}(a,b) < 0$, then (a,b) is a local maxima.

We can combine all of the above steps into one routine as follows:

```
ClassifyStationaryPoints[f_, {x_, y_}] :=
    Module[{X, P, H, g, d, S},
        X = {x, y};
        P = Solve[Grad[f, X] == 0, X, Reals];
        H = Grad[Grad[f, X], X];
        g = H[[1, 1]];
        d = Det[H];
        S[d_, g_] := If[d < 0, "saddle", If[g > 0, "minimum", "maximum"]];

    TableForm[{x, y, d, g, S[d, g], f} /. P,
        TableHeadings -> {None, {x, y, "D", "f_xx", "Type", "f"}}]]
```

Note in particular the `If` command in there that does the second derivative test and the `TableForm` command that writes the results nicely in a table. Read the documentation for the commands at home.

Execute above routine.

Question 5

(a) Type and execute

```
ClassifyStationaryPoints[4 x y - x^3 y - x y^3, {x, y}]
```

to find and classify all the stationary points of $u(x, y) = 4xy - x^3y - xy^3$.

(b) Draw a contour plot to confirm your answer.

Question 6

(a) Use the routine to find and classify all stationary points of $v(x, y) = 5(y^2 - x^2)e^{-x^2-y^2}$.

(b) Draw a contour plot to confirm your answer.